WBEM-SQL1 BNF:

start ::= select-statement
select-statement ::=  SELECT select-list select-expression
select-list ::=  “*”  | property-list 

property-list ::= property-name

)… ]
property-name

 [ ( “,” 
select-expression ::= from-clause [ where-clause ] 

from-clause ::= FROM class-name
where-clause ::= WHERE search-condition
search-condition ::= boolean-term | search-condition OR boolean-term
boolean-term ::= boolean-factor | boolean-term AND boolean-factor
boolean-factor ::=  [ NOT ] boolean-test
boolean-test ::= boolean-primary [ IS [ NOT ] truth-value ] 

boolean-primary ::= predicate | “(” search-condition “)”

predicate ::= comparison-predicate | null-predicate
comparison-predicate ::=  comparison-term comp-op comparison-term
comp-op ::= “=” | “<>” | “<” | “>” | “<=” | “>=” 

null-predicate ::=  comparison-term IS  [ NOT ] NULL

truth-value ::= TRUE | FALSE

property-name = identifier
class-name ::= identifier
comparison-term ::= property-name | numeric-value-expression | string-value-expression | truth-value 

numeric-value-expression ::= [sign] unsigned-numeric-literal
unsigned-numeric-literal ::= exact-numeric-literal | approximate-numeric-literal
exact-numeric-literal ::= unsigned-integer[“.”[unsigned-integer]] | “.”unsigned-integer 

unsigned-integer ::= digit…

approximate-numeric-literal ::= mantissa”E”exponent  

mantissa ::= exact-numeric-literal 

exponent ::= signed-integer 

signed-integer ::= [sign]unsigned-integer
string-value-expression ::= character-string-literal | hex-string-literal | bit-string-literal
character-string-literal ::= quote[(UTF8-CHAR)…] quote
bit-string-literal ::= Bquote[bit... ]quote[ (quote[bit... ]quote)...]

hex-string-literal ::= Xquote[hexit... ]quote[ (quote[hexit... ]quote)... ]

identifier ::= identifier-start [ (identifier-subsequent)… ]

identifier-start ::= upper-case-letter | lower-case-letter | “_”

identifier-subsequent ::= identifier-start | digit
hexit ::= digit | “A”..”F” | “a”..”f”

upper-case-letter ::= “A”..”Z”

lower-case-letter ::= “a”..“z”

digit ::= “0”..“9”

sign ::= “+” | “-”

quote ::= “””

bit ::= “0” | “1”

WBEM-SQL1 YACC GRAMMAR:

%{

%}

%union 

{

   int intValue;

   double doubleValue;

   char* strValue;

   void* nodeValue;

}

%token <intValue> EXACT_NUMBER

%token <intValue> APPROX_NUMBER

%token <strValue> STRING_LITERAL

%token <intValue> TRUE

%token <intValue> FALSE

%token <intValue> NULL

%token <intValue> EQ

%token <intValue> NE

%token <intValue> LT

%token <intValue> LE

%token <intValue> GT

%token <intValue> GE

%token <intValue> NOT

%token <intValue> OR

%token <intValue> AND

%token <intValue> IS

%token <strValue> IDENTIFIER

%token <intValue> SELECT

%token <intValue> WHERE

%token <intValue> FROM

%token <strValue> UNEXPECTED_CHAR

%type <nodeValue> propertyName

%type <nodeValue> propertyList

%type <nodeValue> predicate

%type <nodeValue> comparisonPredicate

%type <nodeValue> comparisonTerm

%type <nodeValue> nullPredicate

%type <nodeValue> searchCondition

%type <nodeValue> fromClause

%type <nodeValue> whereClause

%type <nodeValue> selectStatement

%type <nodeValue> selectList

%type <nodeValue> selectExpression

%type <strValue> className

%left OR

%left AND

%nonassoc NOT

%%

start

    : selectStatement ;

selectStatement

    : SELECT selectList selectExpression ;

selectList

    : '*' ;

    | propertyList ;

propertyList 

    : propertyName ;

    | propertyList ',' propertyName ;

selectExpression 

    : fromClause whereClause ;

    | fromClause ;

fromClause

    : FROM className ;

whereClause 

    : WHERE searchCondition ;

searchCondition 

    : searchCondition OR searchCondition ;

    | searchCondition AND searchCondition ;

    | NOT searchCondition ;

    | '(' searchCondition ')' ;

    | predicate ;

    | predicate IS truthValue ;

    | predicate IS NOT truthValue ;

predicate

    : comparisonPredicate ;

    | nullPredicate ;

comparisonPredicate

    : comparisonTerm LT comparisonTerm ;

    | comparisonTerm GT comparisonTerm ;

    | comparisonTerm LE comparisonTerm ;

    | comparisonTerm GE comparisonTerm ;

    | comparisonTerm EQ comparisonTerm ;

    | comparisonTerm NE comparisonTerm ;

nullPredicate

    : comparisonTerm IS NULL ;

    | comparisonTerm IS NOT NULL ;

truthValue 

    : TRUE ; 

    | FALSE ;

propertyName 

    : IDENTIFIER ;

className 

    : IDENTIFIER ;

comparisonTerm

    : propertyName ;

    | EXACT_NUMBER ;

    | APPROX_NUMBER ;

    | STRING_LITERAL ;

    | truthValue ;

%%
